

IdRel

A package for Identities among Relators

Version 2.03

October 2007

Anne Heyworth
Chris Wensley

Anne Heyworth — Email: anne.heyworth@gmail.com

Chris Wensley — Email: c.d.wensley@bangor.ac.uk
— Homepage: <http://www.informatics.bangor.ac.uk/~cwensley/>
— Address: School of Computer Science, Bangor University,
Dean Street, Bangor, Gwynedd, LL57 1UT, U.K.

Abstract

The IdRel package was originally implemented in 1999, using the GAP 3 language, when the first author was studying for a Ph.D. in Bangor.

This package is designed to compute a minimal set of generators for the module of the identities among relations of a group presentation. It does this using

- rewriting and logged rewriting: a self-contained implementation of the Knuth-Bendix process using the monoid presentation associated to the group presentation;
- monoid polynomials: an implementation of the monoid ring;
- module polynomials: an implementation of the right module over this monoid generated by the relators.
- Y-sequences: used as a *rewriting* way of representing elements of a free crossed module (products of conjugates of group relators and inverse relators).

Bug reports, suggestions and comments are, of course, welcome. Please contact the second author at c.d.wensley@bangor.ac.uk.

Copyright

© 2005-2007 by Anne Heyworth and Chris Wensley

We adopt the copyright regulations of GAP as detailed in the copyright notice in the GAP manual.

Contents

| | | |
|----------|--|-----------|
| 1 | Introduction | 5 |
| 2 | Rewriting Systems | 6 |
| 2.1 | Identity Y-sequences | 6 |
| 2.2 | Monoid Presentations of FpGroups | 7 |
| 2.2.1 | FreeRelatorGroup | 7 |
| 2.2.2 | MonoidPresentationFpGroup | 7 |
| 2.3 | Rewriting systems for FpGroups | 8 |
| 2.3.1 | RewritingSystemFpGroup | 8 |
| 2.3.2 | OnePassReduceWord | 9 |
| 2.3.3 | OnePassKB | 10 |
| 2.4 | Enumerating elements | 11 |
| 2.4.1 | ElementsOfMonoidPresentation | 11 |
| 3 | Logged Rewriting Systems | 12 |
| 3.1 | Logged Knuth-Bendix Completion | 12 |
| 3.1.1 | LoggedOnePassKB | 12 |
| 3.1.2 | LoggedKnuthBendix | 13 |
| 3.2 | Logged reduction of a word | 14 |
| 3.2.1 | LoggedReduceWordKB | 14 |
| 3.2.2 | LoggedRewritingSystemFpGroup | 15 |
| 4 | Monoid Polynomials | 16 |
| 4.1 | Construction of monoid polynomials | 16 |
| 4.1.1 | MonoidPolyFromCoeffsWords | 16 |
| 4.2 | Components of a polynomial | 17 |
| 4.2.1 | Terms | 17 |
| 4.2.2 | Monic | 17 |
| 4.2.3 | AddTermMonoidPoly | 18 |
| 4.3 | Monoid Polynomial Operations | 18 |
| 4.3.1 | Length | 19 |
| 4.4 | Reduction of a Monoid Polynomial | 19 |
| 4.4.1 | ReduceMonoidPoly | 19 |

| | | |
|----------|--|-----------|
| 5 | Module Polynomials | 21 |
| 5.1 | Construction of module polynomials | 21 |
| 5.1.1 | ModulePoly | 21 |
| 5.2 | Components of a module polynomial | 22 |
| 5.2.1 | Terms | 22 |
| 5.2.2 | ZeroModulePoly | 23 |
| 5.2.3 | AddTermModulePoly | 23 |
| 5.3 | Module Polynomial Operations | 23 |
| 5.4 | Identities among relators | 24 |
| 5.4.1 | IdentitiesAmongRelators | 24 |
| 5.4.2 | RootIdentities | 25 |

Chapter 1

Introduction

This manual describes the `IdRel` (version 2.03) GAP package for computing the identities among relations of a group presentation using rewriting, logged rewriting, monoid polynomials, module polynomials and *Y*-sequences.

The theoretical background for these computations is contained in Brown and Huebschmann [BH82], Brown and Razak Salleh [BRS99] and is surveyed in [Hey99].

`IdRel` is primarily designed for the computation of a minimal set of generators for the module of identities among relations. It also contains functions which compute logged rewrite systems for group presentations (and complete them where possible), functions for operations involving elements of monoid rings and functions for operations with elements of right modules over monoid rings. The *Y*-sequences are used as a *rewriting* way of representing elements of a free crossed module (products of conjugates of group relators and inverse relators). The package is written entirely in GAP4, and requires no compilation.

The package is loaded into GAP with the `LoadPackage` command, and on-line help is available in the usual way.

Example

```
gap> LoadPackage( "idrel" );
gap> ?idrel
```

A pdf version of the `IdRel` manual is available in the `doc` directory of the home directory of `IdRel`. The information parameter `InfoIdRel` has default value 0. When raised to a higher value, additional information is printed out. `IdRel` was originally developed in 1999 using GAP3, partially supported by a University of Wales Research Assistantship for the first author, Anne Heyworth. If you use `IdRel` to solve a problem then please send a short email to the second author, to whom bug reports, suggestions and other comments should also be sent. You may reference the package by mentioning [HW03] and [Hey99].

Chapter 2

Rewriting Systems

This chapter describes functions to construct rewriting systems for finitely presented groups which store rewriting information. The example used is a presentation of the quaternion group q_8 with generators a, b and relators $[a^4, b^4, abab^{-1}, a^2b^2]$.

2.1 Identity Y-sequences

A typical input for `ldRel` is an fp-group presentation. This requires a free group F on a set of generators and a set of relators R (words in the free group). The module of identities among relations for this presentation has as its elements the Peiffer equivalence classes of all products of conjugates of relators which represent the identity in the free group.

In this package the identities among relations are represented by Y-sequences, which are lists $[[r_1, u_1], \dots, [r_k, u_k]]$ where r_1, \dots, r_k are the group relators or their inverses, and u_1, \dots, u_k are words in the free group F . A Y-sequence is evaluated in F as the product $(u_1^{-1}r_1u_1) \dots (u_k^{-1}r_ku_k)$ and is an identity Y-sequence if it evaluates to the identity in F . An identity Y-sequence represents an identity among the relators of the group presentation. The main function of the package is to produce a set of Y-sequences which generated the module of identities among relations, and further, that this set be minimal in the sense that every element in it is needed to generate the module.

We now give a simple example to illustrate the use of `ldRel`. All the functions used are described in detail in this manual. We compute a reduced set of identities among relations for the presentation of the symmetric group s_3 with generators a, b and relators $[a^3, b^2, (ab)^2]$.

Example

```
gap> F := FreeGroup( 2 );
<free group on the generators [ f1, f2]>
gap> a := F.1;; b:= F.2;;
gap> rels := [ a^3 , b^2, a*b*a*b];
[ f1^3 , f2^2, f1*f2*f1*f2]
gap> s3 := F/rels;
<fp group on the generators [ f1, f2 ]>
gap> idrels3 := IdentitiesAmongRelators( s3 );;
gap> Display( idrels3 );
[ [ ( FY4*( <identity ...>, FR1*( mon1 - <identity ...> ),
    ( FY8*( <identity ...>, FR2*( mon2 - <identity ...> ),
    ( FY7*( mon2*mon1), FR3*( mon2 - mon1) ),
    ( FY6*( -<identity ...>, FR1*( -mon2*mon1 - mon1) + FR2*( -mon1*mon2 - mo\
```

```

nl - <identity ...> + FR3 *( mon3 + mon2 + <identity ...> ) ],
[ ( FY4*( <identity ...>, FR1*( mon1 - <identity ...> ),
  ( FY8*( <identity ...>, FR2*( mon2 - <identity ...> ),
    ( FY7*( mon2*mon1), FR3 *( mon2 - mon1 ) ),
    ( FY4*( -mon2*mon1 + <identity ...> + FY6*( -<identity ...> + FY7*( -mon\
2*mon1) + FY8*( mon3 ) , FR1*( -mon2 - <identity ...> + FR2*( -mon3 - mon1 - <i
dentity ...> + FR3*( mon3 + mon1 + <identity ...>) ) ] ]

```

2.2 Monoid Presentations of FpGroups

2.2.1 FreeRelatorGroup

◇ `FreeRelatorGroup(grp)` (attribute)
 ◇ `FreeRelatorHomomorphism(grp)` (attribute)

The function `FreeRelatorGroup` returns a free group on the set of relators of the given fp-group *G*. If `HasName(G)` is true then a name is automatically assigned to the free group.

The function `FreeRelatorHomomorphism` returns the group homomorphism from the free group on the relators to the free group on the generators of *G*, mapping each generator to the corresponding word.

Example

```

gap> F := FreeGroup( 2 );;
gap> a := F.1;; b:= F.2;;
gap> rels := [ a^4, b^4, a*b*a*b^-1, a^2*b^2];
[ f1^4, f2^4, f1*f2*f1*f2^-1, f1^2*f2^2]
gap> q8 := F/rels;;
gap> SetName( q8, "q8" );
gap> frq8 := FreeRelatorGroup( q8 );
q8_R
gap> GeneratorsOfGroup( frq8 );
[ q8_R1, q8_R2, q8_R3, q8_R4]
gap> frhomq8 := FreeRelatorHomomorphism( q8 );
[ q8_R1, q8_R2, q8_R3, q8_R4] -> [ f1^4, f2^4, f1*f2*f1*f2^-1, f1^2*f2^2]

```

2.2.2 MonoidPresentationFpGroup

◇ `MonoidPresentationFpGroup(grp)` (attribute)
 ◇ `FreeGroupOfPresentation(mon)` (attribute)
 ◇ `GroupRelatorsOfPresentation(mon)` (attribute)
 ◇ `InverseRelatorsOfPresentation(mon)` (attribute)
 ◇ `HomomorphismOfPresentation(mon)` (attribute)

A monoid presentation for a finitely presented group *G* has two monoid generators g^+, g^- for each group generator *g*. The relators of the monoid presentation comprise the group relators, and relators g^+g^- specifying the inverses. The function `MonoidPresentationFpGroup` returns the monoid

presentation derived in this way from an fp-presentation. (Note: this function should always be followed by a double semicolon – `MonoidPresentationFpGroup(G) ; ;` – because an error occurs in attempting to display the results on the screen: the `ElementsFamily` needs to be fixed.)

The function `FreeGroupOfPresentation` returns the free group on the monoid generators.

The function `GroupRelatorsOfPresentation` returns those relators of the monoid which correspond to the relators of the group. All negative powers in the group relators are converted to positive powers of the g^- .

The function `InverseRelatorsOfPresentation` returns relators which specify the inverse pairs of the monoid generators.

The function `HomomorphismOfPresentation` returns the homomorphism from the free group of the monoid presentation to the free group of the group presentation.

In the example below, the four monoid generators a^+, b^+, a^-, b^- are named `q8_M1`, `q8_M2`, `q8_M3`, `q8_M4`.

Example

```
gap> mon := MonoidPresentationFpgroup( q8 ); ;
gap> fgmon := FreeGroupOfPresentation( mon );
<free group on the generators [ q8_M1, q8_M2, q8_M3, q8_M4 ]>
gap> genfgmon := GeneratorsOfGroup( fgmon );
[ q8_M1, q8_M2, q8_M3, q8_M4 ]
gap> gprels := GroupRelatorsOfPresentation( mon );
[ q8_M1^4, q8_M2^4, q8_M1*q8_M2*q8_M1*q8_M4, q8_M1^2*q8_M2^2 ]
gap> invrels := InverseRelatorsOfPresentation( mon );
[ q8_M1*q8_M3, q8_M2*q8_M4, q8_M3*q8_M1, q8_M4*q8_M2 ]
gap> hompres := HomomorphismOfPresentation( mon );
[ q8_M1, q8_M2, q8_M3, q8_M4 ] -> [ f1, f2, f1^-1, f2^-1 ]
```

2.3 Rewriting systems for FpGroups

These functions duplicate the standard Knuth Bendix functions which are available in the GAP library. There are two reasons for this: (1) these functions were first written before the standard functions were available; (2) we require logged versions of the functions, and these are most conveniently extended versions of the non-logged code.

2.3.1 RewritingSystemFpGroup

◇ `RewritingSystemFpGroup(grp)`

(attribute)

This function attempts to return a complete rewrite system for the group G obtained from the monoid presentation `mon`, with a length-lexicographical ordering on the words in `fgmon`, by applying Knuth-Bendix completion. Such a rewrite system can be obtained for all finite groups. The rewrite rules are (partially) ordered, starting with the inverse relators, followed by the rules which reduce the word length the most.

In our `q8` example there are 16 rewrite rules.

Example

```
gap> rws := RewritingSystemFpGroup( q8 );
```



```
[ [q8_M1*q8_M3, <identity ...>], [ q8_M2*q8_M4, <identity ...> ],
  [q8_M3*q8_M1, <identity ...>], [ q8_M4*q8_M2, <identity ...> ],
  [q8_M1^2*q8_M4, q8_M2], [q8_M1^2*q8_M2, q8_M4], [ q8_M1^3, q8_M3 ],
  [ q8_M4^2, q8_M1^2], [ q8_M4*q8_M3, q8_M1*q8_M4],
  [ q8_M4*q8_M1, q8_M1*q8_M2], [q8_M3*q8_M4, q8_M1*q8_M2],
  [ q8_M3^2, q8_M1^2], [q8_M3*q8_M2, q8_M1*q8_M4],
  [ q8_M2*q8_M3, q8_M1*q8_M2], [q8_M2^2, q8_M1^2],
  [ q8_M2*q8_M1, q8_M1*q8_M4] ]
```

The functions called by `RewritingSystemFpGroup` are as follows.

2.3.2 OnePassReduceWord

◇ `OnePassReduceWord(word, rules)`

(operation)

◇ `ReduceWordKB(word, rules)`

(operation)

Assuming that `word` is an element of a free monoid and `rules` is a list of ordered pairs of such words, the function `OnePassReduceWord` searches the list of rules until it finds that the left-hand side of a rule is a subword of `word`, whereupon it replaces that subword with the right-hand side of the matching rule. The search is continued from the next rule in `rules`, but using the new word. When the end of `rules` is reached, one pass is considered to have been made and the reduced word is returned. If no matches are found then the original word is returned.

The function `ReduceWordKB` repeatedly applies the function `OnePassReduceWord` until the word remaining contains no left-hand side of a rule as a subword. If `rules` is a complete rewrite system, then the irreducible word that is returned is unique, otherwise the order of the rules in `rules` will determine which irreducible word is returned.

Example

```
gap> monrels := Concatenation( gprels, invrels );
[ q8_M1^4, q8_M2^4, q8_M1*q8_M2*q8_M1*q8_M4, q8_M1^2*q8_M2^2, q8_M1*q8_M3,
  q8_M2*q8_M4, q8_M3*q8_M1, q8_M4*q8_M2]
gap> id := One( monrels[1] );
gap> r0 := List( monrels, r -> [ r, id ] );
[ [ q8_M1^4, <identity ...> ], [ q8_M2^4, <identity. ...> ],
  [ q8_M1*q8_M2*q8_M1*q8_M4, <identity ...> ],
  [ q8_M1^2*q8_M2^2, <identity. ...> ], [ q8_M1*q8_M3, <identity ...> ],
  [ q8_M2*q8_M4, <identity ...> ], [ q8_M3*q8_M1, <identity. ...> ],
  [ q8_M4*q8_M2, <identity ...> ] ]
gap> ap := genfgmon[1]; bp := genfgmon[2]; ## p for plus
gap> am := genfgmon[3]; bm := genfgmon[4]; ## m for minus
gap> w0 := (ap^3 * bp^3)^3;
q8_M1^3*q8_M2^3*q8_M1^3*q8_M2^3*q8_M1^3*q8_M2^3
gap> w1 := OnePassReduceWord( w0, r0 );
q8_M1*q8_M2*q8_M1^3*q8_M2^3*q8_M1^3*q8_M2^3
gap> w2 := ReduceWordKB( w0, r0 );
q8_M1*q8_M2*q8_M1*q8_M2*q8_M1*q8_M2
```

2.3.3 OnePassKB

| | |
|--|-------------|
| ◇ OnePassKB(<i>rules</i>) | (operation) |
| ◇ RewriteReduce(<i>rules</i>) | (operation) |
| ◇ KnuthBendix(<i>rules</i>) | (operation) |
| ◇ ShorterRule(<i>rule1</i> , <i>rule2</i>) | (operation) |

The function `OnePassKB` implements the main loop of the Knuth-Bendix completion algorithm. Rules are compared with each other; all critical pairs are calculated; and the irreducible critical pairs are orientated with respect to the length-lexicographical ordering and added to the rewrite system.

The function `RewriteReduce` will remove unnecessary rules from a rewrite system. A rule is deemed to be unnecessary if it is implied by the other rules, i.e. if both sides can be reduced to the same thing by the remaining rules.

The function `KnuthBendix` implements the Knuth-Bendix algorithm, attempting to complete a rewrite system with respect to a length-lexicographic ordering. It calls first `OnePassKB`, which adds rules, and then (for efficiency) `RewriteReduce` which removes any unnecessary ones. This procedure is repeated until `OnePassKB` adds no more rules. It will not always terminate, but for many examples (all finite groups) it will be successful. The rewrite system returned is complete, that is: it will rewrite any given word in the free monoid to a unique irreducible; there is one irreducible for each element of the quotient monoid; and any two elements of the free monoid which are in the same class will rewrite to the same irreducible.

The function `ShorterRule` gives an ordering on rules. Rules $(g_1 g_2, id)$ that identify two generators (or one generator with the inverse of another) are considered to be the best. Otherwise one rule is considered to be better than another if it reduces the length of a word by a greater amount.

One pass of this procedure for our q_8 example adds 13 relators to the original 8, and these 21 are then reduced to 9. A second pass and reduction gives a list of 16 rules which forms a complete rewrite system for the group.

Example

```
gap> r1 := OnePassKB( r0 );
[ [ q8_M1^4, <identity ...> ], [ q8_M2^4, <identity ...> ],
  [ q8_M1*q8_M2*q8_M1*q8_M4, <identity ...> ],
  [ q8_M1^2*q8_M2^2, <identity ...> ], [ q8_M1*q8_M3, <identity ...> ],
  [ q8_M2*q8_M4, <identity ...> ], [ q8_M3*q8_M1, <identity ...> ],
  [ q8_M4*q8_M2, <identity ...> ], [ q8_M2*q8_M1*q8_M4, q8_M1^3 ],
  [ q8_M1*q8_M2^2, q8_M1^3 ], [ q8_M2^2, q8_M1^2 ], [ q8_M1^3, q8_M3 ],
  [ q8_M2^3, q8_M4 ], [ q8_M1*q8_M2*q8_M1, q8_M2 ],
  [ q8_M2^3, q8_M1^2*q8_M2 ], [ q8_M2^2, q8_M1^2 ], [ q8_M1^2*q8_M2, q8_M4 ],
  [ q8_M1^3, q8_M3 ], [ q8_M2*q8_M1*q8_M4, q8_M3 ], [ q8_M1*q8_M2^2, q8_M3 ],
  [ q8_M2^3, q8_M4 ] ]
gap> r1 := RewriteReduce( r1 );
[ [ q8_M1*q8_M3, <identity ...> ], [ q8_M2^2, q8_M1^2 ],
  [ q8_M2*q8_M4, <identity ...> ], [ q8_M3*q8_M1, <identity ...> ],
  [ q8_M4*q8_M2, <identity ...> ], [ q8_M1^3, q8_M3 ],
  [ q8_M1^2*q8_M2, q8_M4 ], [ q8_M1*q8_M2*q8_M1, q8_M2 ],
  [ q8_M2*q8_M1*q8_M4, q8_M3 ] ]
gap> Length( r1 );
9
gap> r2 := KnuthBendix( r1 );
[ [ q8_M1*q8_M3, <identity ...> ], [ q8_M2*q8_M1, q8_M1*q8_M4 ],
```

```
[ q8_M2^2, q8_M1^2], [ q8_M2*q8_M3, q8_M1*q8_M2 ],
[ q8_M2*q8_M4, <identity ...> ], [ q8_M3*q8_M1, <identity ...> ],
[ q8_M3*q8_M2, q8_M1*q8_M4 ], [ q8_M3^2, q8_M1^2 ],
[ q8_M3*q8_M4, q8_M1*q8_M2 ], [ q8_M4*q8_M1, q8_M1*q8_M2 ],
[ q8_M4*q8_M2, <identity ...> ], [ q8_M4*q8_M3, q8_M1*q8_M4 ],
[ q8_M4^2, q8_M1^2], [ q8_M1^3, q8_M3 ], [q8_M1^2*q8_M2, q8_M4 ],
[ q8_M1^2*q8_M4, q8_M2 ] ]
```

2.4 Enumerating elements

2.4.1 ElementsOfMonoidPresentation

◇ ElementsOfMonoidPresentation(*mon*)

(attribute)

The function `ElementsOfMonoidPresentation` returns a list of normal forms for the elements of the group given by the monoid presentation `mon`. The normal forms are in fact the least elements in each equivalence class (with respect to the length-lex order). The function `EnumerateKB` builds up a catalogue of irreducible words in the generators of a monoid with respect to a set of rules. When `rules` is a complete rewrite system for G the list returned is a set of normal forms for the group elements.

Example

```
gap> elq8 := Elements( q8 );
[ <identity .. .>, f1, f2, f1^2, f1*f2, f1^3*f2, f1^3, f1^2*f2 ]
gap> elmonq8 := ElementsOfMonoidPresentation( monq8 );
[ <identity. ...>, q8_M1, q8_M2, q8_M3, q8_M4, q8_M1^2, q8_M1*q8_M2,
  q8_M1*q8_M4 ]
```

Chapter 3

Logged Rewriting Systems

A logged rewrite system is associated with a group presentation. Each logged rewrite rule contains, in addition to the standard rewrite rule, a record or log component which express the rule in terms of the original relators of the group. Here a logged rewrite rule is represented by a triple $[u, [L_1, L_2, \dots, L_k], v]$, where $[u, v]$ is a rewrite rule and $L_i = [n_i, w_i]$ where n_i is a group relator and w_i is a word. These three components obey the identity $u = n_1^{w_1} \dots n_k^{w_k} v$.

Rules of the form $g^+ g^-$ are relevant to the monoid presentation, but not to the group presentation, so are not included in the logging.

3.1 Logged Knuth-Bendix Completion

The functions in this section are the logged versions of those in the previous chapter.

3.1.1 LoggedOnePassKB

◇ `LoggedOnePassKB(loggedrules)` (operation)

Given a logged rewrite system, this function finds all the rules that would be added to complete the rewrite system in `OnePassKB`, and also the logs which relate the new rules to the originals. The result of applying this function to `loggedrules` is to add new logged rules to the system without changing the monoid it defines.

Example

```
gap> l0 := ListWithIdenticalEntries( 8, 0 );;
gap> for j in [1..8] do
  r := r0[j];
  if ( j<5 ) then
    l0[j] := [ r[1], [ [j,id] ], r[2] ];
  else
    l0[j] := [ r[1], [ ], r[2] ];
  fi;
od;
gap> l0;
[ [ q8_M1^4, [ [ 1, <identity ...> ] ], <identity. ...> ],
  [ q8_M2^4, [ [ 2, <identity ...> ] ], <identity ...> ],
  [ q8_M1*q8_M2*q8_M1*q8_M4, [ [ 3, <identity ...> ] ], <identity ...> ],
```

```

[ q8_M1^2*q8_M2^2, [ [ 4, <identity ...> ] ], <identity ...> ],
[ q8_M1*q8_M3, [ ], <identity ...> ],
[ q8_M2*q8_M4, [ ], <identity ...> ],
[ q8_M3*q8_M1, [ ], <identity ...> ],
[ q8_M4*q8_M2, [ ], <identity ...> ] ]
gap> l1 := LoggedOnePassKB( l0 );
[ [ q8_M1^4, [ [ 1, <identity. ...> ] ], <identity ...> ],
  [ q8_M2^4, [ [ 2, <identity ...> ] ], <identity ...> ],
  [ q8_M1*q8_M2*q8_M1*q8_M4, [ [ 3, <identity ...> ] ], <identity ...> ],
  [ q8_M1^2*q8_M2^2, [ [ 4, <identity ...> ] ], <identity ...> ],
  [ q8_M1*q8_M3, [ ], <identity ...> ], [ q8_M2*q8_M4, [ ], <identity ...> ],
  [ q8_M3*q8_M1, [ ], <identity ...> ], [ q8_M4*q8_M2, [ ], <identity ...> ],
  [ q8_M2*q8_M1*q8_M4, [ [ -1, <identity. ...> ], [ 3, q8_M1^-3 ] ], q8_M1^3],
  [ q8_M1*q8_M2^2, [ [ -1, <identity ...> ], [ 4, q8_M1^-3 ] ], q8_M1^3],
  [ q8_M2^2, [ [ -1, <identity ...> ], [ 4, q8_M1^-2 ] ], q8_M1^2],
  [ q8_M1^3, [ [ 1, <identity ...> ] ], q8_M3],
  [ q8_M2^3, [ [ 2, <identity ...> ] ], q8_M4],
  [ q8_M1*q8_M2*q8_M1, [ [ 3, <identity ...> ] ], q8_M2 ],
  [ q8_M2^3, [ [ -4, <identity ...> ], [ 2, q8_M2^-1*q8_M1^-2 ] ],
    q8_M1^2*q8_M2],
  [ q8_M2^2, [ [ -4, <identity ...> ], [ 2, q8_M1^-2 ] ], q8_M1^2],
  [ q8_M1^2*q8_M2, [ [ 4, <identity ...> ] ], q8_M4],
  [ q8_M1^3, [ [ 1, q8_M3^-1 ] ], q8_M3],
  [ q8_M2*q8_M1*q8_M4, [ [ 3, q8_M3^-1 ] ], q8_M3 ],
  [ q8_M1*q8_M2^2, [ [ 4, q8_M3^-1 ] ], q8_M3 ],
  [ q8_M2^3, [ [ 2, q8_M4^-1 ] ], q8_M4] ]

```

3.1.2 LoggedKnuthBendix

◇ LoggedKnuthBendix(*loggedrules*)

(operation)

◇ LoggedRewriteReduce(*loggedrules*)

(operation)

The function `LoggedRewriteReduce` removes unnecessary rules from a logged rewrite system. It works on the same principle as `RewriteReduce`.

The function `LoggedKnuthBendix` repeatedly applies `LoggedOnePassKB` and `LoggedRewriteReduce` until no new rules are added and no unnecessary ones are included. The output is a reduced complete logged rewrite system.

Example

```

gap> l1 := LoggedRewriteReduce( l1 );
[ [ q8_M1*q8_M3, [ ], <identity ...> ],
  [ q8_M2^2, [ [ -4, <identity ...> ], [ 2, q8_M1^-2 ] ], q8_M1^2 ],
  [ q8_M2*q8_M4, [ ], <identity ...> ], [ q8_M3*q8_M1, [ ], <identity ...> ],
    [ q8_M4*q8_M2, [ ], <identity ...> ],
  [ q8_M1^3, [ [ 1, <identity. ...> ] ], q8_M3 ],
  [ q8_M1^2*q8_M2, [ [ 4, <identity. ...> ] ], q8_M4 ],
  [ q8_M1*q8_M2*q8_M1, [ [ 3, <identity ...> ] ], q8_M2 ],
  [ q8_M2*q8_M1*q8_M4, [ [ 3, q8_M3^-1 ] ], q8_M3] ]
gap> l2 := LoggedKnuthBendix( l1 );
[ [ q8_M1*q8_M3, [ ], <identity ...> ],

```

```

[ q8_M2*q8_M1, [ [ 3, q8_M3^-1 ], [-1, <identity ...> ], [ 4, q8_M1^-1 ] ],
  q8_M1*q8_M4 ],
[ q8_M2^2, [ [ -4, <identity ...> ], [2, q8_M1^-2] ], q8_M1^2 ],
[ q8_M2*q8_M3, [ [ -3, <identity ...> ] ], q8_M1*q8_M2 ],
[ q8_M2*q8_M4, [ ], <identity ...> ], [ q8_M3*q8_M1, [ ], <identity ...> ],
[ q8_M3*q8_M2, [ [ -1, <identity ...> ], [4, q8_M1^-1 ] ], q8_M1*q8_M4 ],
[ q8_M3^2, [ [ -1, <identity ...> ] ], q8_M1^2 ],
[ q8_M3*q8_M4, [ [ -1, <identity ...> ], [ -2, q8_M1^-2 ],
  [ 4, <identity ...> ], [ 3, q8_M3^-1*q8_M2^-1 ],
  [ -3, <identity ...> ] ], q8_M1*q8_M2 ],
[ q8_M4*q8_M1, [ [ -4, <identity ...> ], [ 3, q8_M1^-1 ] ], q8_M1*q8_M2 ],
[ q8_M4*q8_M2, [ ], <identity ...> ],
[ q8_M4*q8_M3, [ [ -3, q8_M3^-1*q8_M4^-1 ] ], q8_M1*q8_M4 ],
[ q8_M4^2, [ [ -4, <identity ...> ] ], q8_M1^2 ],
[ q8_M1^3, [ [ 1, <identity ...> ] ], q8_M3 ],
[ q8_M1^2*q8_M2, [ [4, <identity ...> ] ], q8_M4 ],
[ q8_M1^2*q8_M4, [ [ -4, q8_M1^-2 ], [ 1, <identity ...> ] ], q8_M2 ] ]

```

3.2 Logged reduction of a word

3.2.1 LoggedReduceWordKB

◇ `LoggedReduceWordKB(loggedrules)` (operation)
 ◇ `LoggedOnePassReduceWord(word, loggedrules)` (operation)
 ◇ `ShorterLoggedRule(logrule1, logrule2)` (operation)

Given a word and a logged rewrite system, the function `LoggedOnePassReduceWord` makes one reduction of the word (as in `OnePassReduceWord`) and records this, using the log part of the rule used and the position in the original word of the replaced part.

The function `LoggedReduceWordKB` repeatedly applies `OnePassLoggedReduceWord` until the word can no longer be reduced. Each step of the reduction is logged, showing how the original word can be expressed in terms of the original relators and the irreducible word. When `loggedrules` is complete the reduced word is a unique normal form for that group element. The log of the reduction however has no uniqueness but depends on the order in which the rules are applied and on the non-unique logs of the rules themselves.

The function `Shorterloggedrule` decides whether one logged rule is better than another, using the same criteria as `ShorterRule`.

— Example —

```

gap> w0;
q8_M1^3*q8_M2^3*q8_M1^3*q8_M2^3*q8_M1^3*q8_M2^3
gap> lw1 := LoggedOnePassReduceWord( w0, l2 );
[ [ [ 3, q8_M3^-1*q8_M2^-2*q8_M1^-3 ], [ -1, q8_M2^-2*q8_M1^-3 ],
  [ 4, q8_M1^-1*q8_M2^-2*q8_M1^-3 ], [ -4, q8_M1^-3 ], [ 2, q8_M1^-5 ],
  [ -4, q8_M1^-6 ], [ 3, q8_M1^-7 ], [ 1, <identity ...> ],
  [ 4, q8_M1^-2*q8_M3^-1 ], [ -4, q8_M1^-2*q8_M3^-1 ], [ 1, q8_M3^-1 ] ],
  q8_M3*q8_M2*q8_M1*q8_M2^3*q8_M1^3*q8_M2^3 ]
gap> lw2 := LoggedReduceWordKB( w0, l2 );
[ [ [ 3, q8_M3^-1*q8_M2^-2*q8_M1^-3 ], [ -1, q8_M2^-2*q8_M1^-3 ],

```

```

[ 4, q8_M1^-1*q8_M2^-2*q8_M1^-3 ], [ -4, q8_M1^-3 ], [ 2, q8_M1^-5 ],
[ -4, q8_M1^-6 ], [ 3, q8_M1^-7 ], [ 1, <identity ...> ],
[ 4, q8_M1^-2*q8_M3^-1 ], [ -4, q8_M1^-2*q8_M3^-1 ], [ 1, q8_M3^-1 ],
[ 3, q8_M3^-2 ], [ -1, q8_M3^-1 ], [ 4, q8_M1^-1*q8_M3^-1 ],
[ -4, q8_M4^-1*q8_M1^-1*q8_M3^-1 ],
[ 2, q8_M1^-2*q8_M4^-1*q8_M1^-1*q8_M3^-1 ], [ -4, <identity ...> ],
[ 3, q8_M1^-1 ], [ 1, q8_M2^-1*q8_M1^-1*q8_M2^-1*q8_M1^-1 ],
[ 3, q8_M3^-1*q8_M1^-1 ], [ -1, q8_M1^-1 ], [ 4, q8_M1^-2 ],
[ -4, q8_M3^-1*q8_M2^-1*q8_M4^-1*q8_M1^-2 ],
[ 2, q8_M1^-2*q8_M3^-1*q8_M2^-1*q8_M4^-1*q8_M1^-2 ],
[ -3, q8_M4^-1*q8_M1^-2 ], [ -4, q8_M1^-2 ], [ 3, q8_M1^-3 ],
[ 1, <identity ...> ], [ 4, q8_M2^-2*q8_M3^-1 ], [ -4, q8_M3^-1 ],
[ 2, q8_M1^-2*q8_M3^-1 ] ], q8_M1*q8_M4 ]

```

3.2.2 LoggedRewritingSystemFpGroup

◇ LoggedRewritingSystemFpGroup(*loggedrules*)

(attribute)

Given a group presentation, the function `LoggedRewritingSystemFpGroup` determines a logged rewrite system based on the relators. The initial logged rewrite system associated with a group presentation consists of two types of rule. These are logged versions of the two types of rule in the monoid presentation. For each relator `rel` of the group there is a logged rule `[rel, [[1, rel]], id]`. For each inverse relator there is a logged rule `[gen*inv, [], id]`. It then attempts a completion of the logged rewrite system. The rules in the final system are partially ordered by the function `ShorterLoggedRule`.

Example

```

gap> LoggedRewritingSystemFpGroup( q8 );
[ [ q8_M4*q8_M2, [ ], <identity ...> ], [ q8_M3*q8_M1, [ ], <identity ...> ],
  [ q8_M2*q8_M4, [ ], <identity ...> ],
  [ q8_M1*q8_M3, [ ], <identity ...> ],
  [ q8_M1^2*q8_M4, [ [ -8, q8_M1^-2 ], [ 5, <identity ...> ] ], q8_M2 ],
  [ q8_M1^2*q8_M2, [ [ 8, <identity ...> ] ], q8_M4 ],
  [ q8_M1^3, [ [ 5, <identity ...> ] ], q8_M3 ],
  [ q8_M4^2, [ [ -8, <identity ...> ] ], q8_M1^2 ],
  [ q8_M4*q8_M3, [ [ -7, q8_M3^-1*q8_M4^-1 ] ], q8_M1*q8_M4 ],
  [ q8_M4*q8_M1, [ [ -8, <identity ...> ], [ 7, q8_M1^-1 ] ], q8_M1*q8_M2 ],
  [ q8_M3*q8_M4,
    [ [ -5, <identity ...> ], [ -6, q8_M1^-2 ], [ 8, <identity ...> ],
      [ 7, q8_M3^-1*q8_M2^-1 ], [ -7, <identity ...> ] ], q8_M1*q8_M2 ],
  [ q8_M3^2, [ [ -5, <identity ...> ] ], q8_M1^2 ],
  [ q8_M3*q8_M2, [ [ -5, <identity ...> ], [ 8, q8_M1^-1 ] ], q8_M1*q8_M4 ],
  [ q8_M2*q8_M3, [ [ -7, <identity ...> ] ], q8_M1*q8_M2 ],
  [ q8_M2^2, [ [ -a, <identity ...> ], [ 6, q8_M1^-2 ] ], q8_M1^2 ],
  [ q8_M2*q8_M1, [ [ 7, q8_M3^-1 ], [ -5, <identity ...> ], [ a, q8_M1^-1 ] ],
    q8_M1*q8_M4 ] ]

```

Chapter 4

Monoid Polynomials

This chapter describes functions to compute with elements of a free noncommutative algebra. The elements of the algebra are sums of rational multiples of words in a free monoid. These will be called monoid polynomials, and are stored as lists of pairs `[coefficient, word]`.

4.1 Construction of monoid polynomials

4.1.1 MonoidPolyFromCoeffsWords

◇ `MonoidPolyFromCoeffsWords(coeffs, words)` (operation)
◇ `MonoidPoly(terms)` (operation)
◇ `ZeroMonoidPoly(F)` (operation)

There are two possible ways to input a monoid polynomial. The first is by listing the coefficients and then the words; the second is by listing the terms as a list of pairs `[coefficient, word]`. Note that if a word occurs more than once in the input list, the coefficients will be added so that the terms of the monoid polynomial recorded do not contain any duplicates. The zero monoid polynomial is the polynomial with no terms.

— Example —

```
gap> rels := RelatorsOfFpGroup( q8 );
[ f1^4, f2^4, f1*f2*f1*f2^-1, f1^2*f2^2 ]
gap> freeq8 := FreeGroupOfFpGroup( q8 );
Group( [f1, f2] )
gap> gens := GeneratorsOfGroup( freeq8 );
gap> famfree := ElementsFamily( FamilyObj( freeq8 ) );
NewFamily( "FreeGroupElementsFamily", [ 125, 144, 989 ],
[ 76, 77, 78, 79, 111, 114, 117, 121, 125, 144, 989 ] )
gap> famfree!.monoidPolyFam := MonoidPolyFam;
NewFamily( "MonoidPolyFam", [ 2416 ], [ 111, 114, 117, 2416 ] )
gap> cg := [6,7];
[ 6, 7 ]
gap> cr := [3,4,-5,-2];
[ 3, 4, -5, -2 ]
gap> pg := MonoidPolyFromCoeffsWords( cg, gens );
<monpoly>
gap> Print( pg );
```



```

7*f2 + 6*f1
gap> pr := MonoidPolyFromCoeffsWords( cr, rels );
<monpoly>
gap> Print( pr );
4*f2^4 - 5*f1*f2*f1*f2^-1 - 2*f1^2*f2^2 + 3*f1^4
gap> zeromp := ZeroMonoidPoly( freeq8 );
<monpoly>
gap> Print( zeromp );
zero monpoly

```

4.2 Components of a polynomial

4.2.1 Terms

- ◇ `Terms(poly)` (attribute)
- ◇ `Coeffs(poly)` (attribute)
- ◇ `Words(poly)` (attribute)
- ◇ `LeadTerm(poly)` (attribute)
- ◇ `LeadCoeffMonoidPoly(poly)` (attribute)

The function `Terms` returns the terms of a polynomial as a list of pairs of the form `[word, coefficient]`. The function `Coeffs` returns the coefficients of a polynomial as a list, and the function `Words` returns the words of a polynomial as a list. The function `LeadTerm` returns the term of the polynomial whose word component is the largest with respect to the length-lexicographical ordering. The function `LeadCoeffMonoidPoly` returns the coefficient of the leading term of a polynomial.

Example

```

gap> Coeffs( pr );
[ 4, -5, -2, 3 ]
gap> Terms( pr );
[ [ 4, f2^4 ], [ -5, f1*f2*f1*f2^-1 ], [ -2, f1^2*f2^2 ], [ 3, f1^4 ] ]
gap> Words( pr );
[ f2^4, f1*f2*f1*f2^-1, f1^2*f2^2, f1^4 ]
gap> LeadTerm( pr );
[ 4, f2^4 ]
gap> LeadCoeffMonoidPoly( pr );
4

```

4.2.2 Monic

- ◇ `Monic(poly)` (operation)

A monoid polynomial is called *monic* if the coefficient of its leading polynomial is one. The function `Monic` converts a polynomial into a monic polynomial by dividing all the coefficients by the leading coefficient.

Example

```
gap> mpr := Monic( pr );;
gap> Print( mpr );
f2^4 - 5/4*f1*f2*f1*f2^-1 - 1/2*f1^2*f2^2 + 3/4*f1^4
```

4.2.3 AddTermMonoidPoly

◇ AddTermMonoidPoly(*poly*, *coeff*, *word*)

(operation)

The function AddTermMonoidPoly adds a new term, given by its coefficient and word, to an existing polynomial.

Example

```
gap> w := gens[1]^gens[2];
f2^-1*f1*f2
gap> cw := 3/4;;
gap> wpg:= AddTermMonoidPoly( pg, cw, w);
<monopoly>
gap> Print( wpg );
3/4*f2^-1*f1*f2 + 7*f2 + 6*f1
```

4.3 Monoid Polynomial Operations

Tests for equality and arithmetic operations are performed in the usual way.

The operation `poly1 = poly2` returns `true` if the monoid polynomials have the same terms, and `false` otherwise. Multiplication of a monoid polynomial (on the left or right) by a coefficient; the addition or subtraction of two monoid polynomials; multiplication (on the right) of a monoid polynomial by a word; and multiplication of two monoid polynomials; are all implemented.

Example

```
gap> pg = pg;
true
gap> pg = pr;
false
gap> prcw := pr*cw;
3*f2^4 - 15/4*f1*f2*f1*f2^-1 - 3/2*f1^2*f2^2 + 9/4*f1^4
gap> cwpr := cw*pr;
3*f2^4 - 15/4*f1*f2*f1*f2^-1 - 3/2*f1^2*f2^2 + 9/4*f1^4
gap> pr = prcw;
false
gap> prcw = cwpr;
true
gap> Print( pg + pr );
4*f2^4 - 5*f1*f2*f1*f2^-1 - 2*f1^2*f2^2 + 3*f1^4 + 7*f2 + 6*f1
gap> Print( pg - pr );
- 4*f2^4 + 5*f1*f2*f1*f2^-1 + 2*f1^2*f2^2 - 3*f1^4 + 7*f2 + 6*f1
```

```
gap> Print( pg * w );
6*f1*f2^-1*f1*f2 + 7*f1*f2
gap> Print( pg * pr );
28*f2^5 - 35*f2*f1*f2*f1*f2^-1 - 14*f2*f1^2*f2^2 + 21*f2*f1^4
+ 24*f1*f2^4 - 30*f1^2*f2*f1*f2^-1 - 12*f1^3*f2^2 + 18*f1^5
```

4.3.1 Length

◇ Length(*poly*)

(attribute)

This function returns the number of distinct terms in the monoid polynomial.

Example

```
gap> Length( pr );
4
```

The boolean function `poly1 > poly2` returns `true` if the first polynomial has more terms than the second. If the polynomials are the same length it will compare their leading terms. If the leading word of the first is lengthlexicographically greater than the leading word of the second, or if the words are equal but the coefficient of the first is greater than the coefficient of the second then `true` is returned. If the leading terms are equal then the next terms are compared in the same way. If all terms are the same then `false` is returned.

Example

```
gap> pol > 3*pol;
false
gap> pol > redpol;
true
```

4.4 Reduction of a Monoid Polynomial

4.4.1 ReduceMonoidPoly

◇ ReduceMonoidPoly(*poly*, *rules*)

(operation)

Recall that the words of a monoid polynomial are elements of a free monoid. Given a rewrite system (set of rules) on the free monoid the words can be reduced. This allows us to simulate calculation in monoid rings where the monoid is given by a complete presentation. This function reduces the words of the polynomial (elements of the free monoid) with respect to the complete rewrite system. The words of the reduced polynomial are normal forms for the elements of the monoid presented by that rewrite system.

Example

```
gap> M := genfgmon;
[ q8_M1, q8_M2, q8_M3, q8_M4 ]
```

```

gap> r2;
[ q8_M1*q8_M3, <identity ...>], [ q8_M2*q8_M1, q8_M1*q8_M4 ],
[ q8_M2^2, q8_M1^2], [ q8_M2*q8_M3, q8_M1*q8_M2 ],
[ q8_M2*q8_M4, <identity ...> ], [ q8_M3*q8_M1, <identity ...> ],
[ q8_M3*q8_M2, q8_M1*q8_M4], [ q8_M3^2, q8_M1^2 ],
[ q8_M3*q8_M4, q8_M1*q8_M2], [ q8_M4*q8_M1, q8_M1*q8_M2 ],
[ q8_M4*q8_M2, <identity ...> ], [ q8_M4*q8_M3, q8_M1*q8_M4 ],
[ q8_M4^2, q8_M1^2 ], [ q8_M1^3, q8_M3], [ q8_M1^2*q8_M2, q8_M4 ],
[ q8_M1^2*q8_M4, q8_M2 ] ]
gap> pol := MonoidPolyFromCoeffsWords([3,2,-5], [M[1]*M[3] ,M[2]-4,M[1]]);
<monopoly>
gap> Print( pol );
2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1
gap> redpol := ReduceMonoidPoly( pol, KBrules );
<monopoly>
gap> Print( redpol );
- 5*q8_M1 + 5*<identity ...>

```

Chapter 5

Module Polynomials

In this chapter we consider finitely generated modules over the monoid rings considered previously. We call an element of this module a *module polynomial*, and we describe functions to construct module polynomials and the standard algebraic operations for such polynomials.

A module polynomial `modpoly` is recorded as a list of pairs, `[gen, monpoly]`, where `gen` is a module generator (basis element), and `monpoly` is a monoid polynomial. The module polynomial is printed as the formal sum of monoid polynomial multiples of the generators. Note that the monoid polynomials are the coefficients of the module polynomials and appear to the right of the generator, as we choose to work with right modules.

The examples we are aiming for are the identities among the relators of a finitely presented group (see section 5.4).

5.1 Construction of module polynomials

5.1.1 ModulePoly

◇ `ModulePoly(gens, monpolys)` (operation)
◇ `ModulePoly(args)` (operation)

This function returns a module polynomial. The terms of the polynomial maybe input as a list of generators followed by a list of monoid polynomials or as one list of `[generator, monoid polynomial]` pairs.

Example

```
gap> frq8 := FreeRelatorGroup( q8 );
q8_R
gap> genfrq8 := GeneratorsOfGroup( frq8 );
[ q8_R1, q8_R2, q8_R3, q8_R4 ]
gap> mp1 := MonoidPolyFromCoeffsWords( [3,2,-5], [M[1]*M[3], M[2]^4, M[1] ] );
<monpoly>
gap> Print( mp1 );
2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1
gap> mp2 := MonoidPolyFromCoeffsWords( [1,-4,-1,2],
[ M[2]*M[3]*M[1], M[3]^3, M[1]*M[2], M[4] ]
<monpoly>
gap> Print( mp2 );
```

```

- 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 - q8_M1*q8_M2 + 2*q8_M4
gap> s1 := ModulePoly( [ genfrq8[4] , genfrq8[1] ], [mp1, mp2 ] );
<modpoly>
gap> Print( s1 );
q8_R1*( - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 - q8_M1*q8_M2 + 2*q8_M4)
+ q8_R4*( 2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1)
gap> s2 := ModulePoly( [ genfrq8[3], genfrq8[2], genfrq8[1] ], [3*mp1, -1*mp2,\
(mp1+mp2) ] );
<modpoly>
gap> Print( s2 );
q8_R1*(2*q8_M2^4 - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 + 3*q8_M1*q8_M3
- q8_M1*q8_M2 + 2*q8_M4 - 5*q8_M1) + q8_R2*(4*q8_M3^3 - q8_M2*q8_M3*q8_M1
+ q8_M1*q8_M2 - 2*q8_M4) + q8_R3*(6*q8_M2^4 + 9*q8_M1*q8_M3 - 15*q8_M1)

```

5.2 Components of a module polynomial

5.2.1 Terms

- ◇ *Terms(modpoly)* (attribute)
- ◇ *LeadTerm(modpoly)* (attribute)
- ◇ *LeadMonoidPoly(modpoly)* (attribute)
- ◇ *One(modpoly)* (attribute)
- ◇ *Length(modpoly)* (attribute)

The first function counts the number of module generators which occur in *modpoly* (a generator occurs in a polynomial if it has nonzero coefficient). The function *One* returns the identity in the free group on the generators.

The function *Terms* returns the terms of a module polynomial as a list of pairs. In *LeadTerm*, the generators are ordered, and the term of *modpoly* with the highest value generator is defined to be the leading term. The monoid polynomial (coefficient) part of the leading term is returned by the function *LeadMonoidPoly*.

Example

```

gap> Length(s1);
2
gap> Length(s2);
3
gap> One( s1 );
<identity ...>
gap> Terms( s1 );
[ [ q8_R1, - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 - q8_M1*q8_M2 + 2*q8_M4 ],
  [ q8_R4, 2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1 ] ]
gap> Terms( s2 );
[ [ q8_R1, 2*q8_M2^4 - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 + 3*q8_M1*q8_M3 - q8_M1*q8_M\
2 + 2*q8_M4 - 5*q8_M1 ],
  [ q8_R2, 4*q8_M3^3 - q8_M2*q8_M3*q8_M1 + q8_M1*q8_M2 - 2*q8_M4 ],
  [ q8_R3, 6*q8_M2^4 + 9*q8_M1*q8_M3 - 15*q8_M1 ] ]
gap> LeadTerm( s1 );
[ q8_R4, <monpoly> ]

```

```
gap> LeadTerm( s2 );
[ q8_R3, <monpoly> ]
gap> Print( LeadMonoidPoly( s1 ) );
2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1
gap> Print( LeadMonoidPoly( s2 ) );
6*q8_M2^4 + 9*q8_M1*q8_M3 - 15*q8_M1
```

5.2.2 ZeroModulePoly

◇ ZeroModulePoly(*Fgens*, *Fmon*)

(operation)

Assuming that *Fgens* is the free group on the module generators and *Fmon* is the free group on the monoid generators, then this function returns the zero module polynomial, which has no terms, and is an element of the module.

Example

```
gap> zeromp := ZeroModulePoly( frel, freeq8 );
<modpoly>
gap> Print( zeromp );
zero modpoly
```

5.2.3 AddTermModulePoly

◇ AddTermModulePoly(*modpoly*, *gen*, *monpoly*)

(operation)

This function adds a term [*gen*, *monpoly*] to a module polynomial *modpoly*.

Example

```
gap> s3 := AddTermModulePoly( s1, frelgen[3], mpl );
<modpoly>
gap> Print( s3 );
q8_R1*( - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 - q8_M1*q8_M2 + 2*q8_M4)
+ q8_R3*( 2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1)
+ q8_R4*( 2*q8_M2^4 + 3*q8_M1*q8_M35*q8_M1)
```

5.3 Module Polynomial Operations

Tests for equality and arithmetic operations are performed in the usual way. Module polynomials may be added or subtracted. A module polynomial can also be multiplied on the right by a word or by a scalar. The effect of this is to multiply the monoid polynomial parts of each term by the word or scalar. This is made clearer in the example.

Example

```
gap> Print( s1 );
q8_R1*( - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 - q8_M1*q8_M2 + 2*q8_M4)
```

```

+ q8_R4*( 2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1)
gap> Print( s2 );
q8_R1*(2*q8_M2^4 - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 + 3*q8_M1*q8_M3
- q8_M1*q8_M2 + 2*q8_M4 - 5*q8_M1)
+ q8_R2*(4*q8_M3^3 - q8_M2*q8_M3*q8_M1 + q8_M1*q8_M2 - 2*q8_M4)
+ q8_R3*(6*q8_M2^4 + 9*q8_M1*q8_M3 - 15*q8_M1)
gap> Print( s1+s2 );
q8_R1*(2*q8_M2^4 - 8*q8_M3^3 + 2*q8_M2*q8_M3*q8_M1
+ 3*q8_M1*q8_M3^2*q8_M1*q8_M2 + 4*q8_M4 - 5*q8_M1)
+ q8_R2*(4*q8_M3^3 - q8_M2*q8_M3*q8_M1 + q8_M1*q8_M2 - 2*q8_M4)
+ q8_R3*(6*q8_M2^4 + 9*q8_M1*q8_M3 - 15*q8_M1)
+ q8_R4*(2*q8_M2^4 + 3*q8_M1*q8_M3 - 5*q8_M1)
gap> Print( s1 - s2 );
q8_R1*( - 2*q8_M2^4 - 3*q8_M1*q8_M3 + 5*q8_M1)
+ q8_R2*( - 4*q8_M3^3 + q8_M2*q8_M3*q8_M1 - q8_M1*q8_M2 + 2*q8_M4)
+ q8_R3*( - 6*q8_M2^4 + 9*q8_M1*q8_M3 - 15*q8_M1) + q8_R4*(2*q8_M2^4
+ 3*q8_M1*q8_M3 - 5*q8_M1)
gap> Print( s1*1/2 );
q8_R1*( - 2*q8_M3^3 + 1/2*q8_M2*q8_M3*q8_M1 - 1/2*q8_M1*q8_M2 + q8_M4)
+ q8_R4*( q8_M2^4 + 3/2*q8_M1*q8_M3 - 5/2*q8_M1)
gap> M[1];
q8_M1
gap> Print( s1*M[1] );
q8_R1*( - 4*q8_M3^3*q8_M1 + q8_M2*q8_M3*q8_M1^2 - q8_M1*q8_M2*q8_M1
+ 2*q8_M4*q8_M1) + q8_R4*( 2*q8_M2^4*q8_M1 + 3*q8_M1*q8_M3*q8_M1 - 5*q8_M1^2)
gap> Length( s3 );
3

```

5.4 Identities among relators

5.4.1 IdentitiesAmongRelators

◇ IdentitiesAmongRelators(*grp*)

(attribute)

The identities among the relators for a finitely presented group are logged module polynomials.

Example

```

gap> idsq8 := IdentitiesAmongRelators( q8 );
[ [ ( q8_Y3*( q8_M1*q8_M4), q8_R1*( q8_M1 - <identity ...> ) ),
    ( q8_Y10*( -q8_M1*q8_M4), q8_R2*( q8_M2 - <identity ...> ) ),
    ( q8_Y17*( <identity ...>), q8_R1*( -q8_M3 - q8_M2) + q8_R3*( q8_M1^
      2 + q8_M3 + q8_M1 + <identity ...> ) ),
    ( q8_Y31*( q8_M1*q8_M4), q8_R3*( q8_M3 - q8_M2) + q8_R4*( q8_M1 - <identity\
      ...> ) ), ( q8_Y32*( -q8_M1*q8_M4), q8_R2*( -q8_M1^
      2) + q8_R3*( -q8_M3 - <identity ...> ) + q8_R4*( q8_M2 + <identity ...> ) )
    ,
    ( q8_Y12*( q8_M1*q8_M4), q8_R1*( -q8_M2) + q8_R3*( q8_M1*q8_M2 + q8_M4) + q\
      8_R4*( q8_M2 - <identity ...> ) ),
    ( q8_Y16*( -<identity ...>), q8_R1*( -<identity ...> ) + q8_R2*( -q8_M1) + q\
      8_R4*( q8_M3 + q8_M1) ) ],

```



```

[ ( q8_Y3*( q8_M1*q8_M4), q8_R1*( q8_M1 - <identity ...> ) ),
  ( q8_Y10*( -q8_M1*q8_M4), q8_R2*( q8_M2 - <identity ...> ) ),
  ( q8_Y3*( -q8_M1*q8_M4*q8_M3) + q8_Y17*( <identity ...>), q8_R1*( -q8_M2 - \
<identity ...>) + q8_R3*( q8_M1^2 + q8_M3 + q8_M1 + <identity ...>) ),
  ( q8_Y31*( q8_M1*q8_M4), q8_R3*( q8_M3 - q8_M2) + q8_R4*( q8_M1 - <identity\
...>) ),
  ( q8_Y10*( -q8_M1*q8_M4*q8_M2 - q8_M1*q8_M4) + q8_Y32*( -q8_M1*q8_M4), q8_R\
2*( -<identity ...>) + q8_R3*( -q8_M3 - <identity ...>) + q8_R4*( q8_M2 + <identi\
ty ...>) ),
  ( q8_Y12*( q8_M1*q8_M4), q8_R1*( -q8_M2) + q8_R3*( q8_M1*q8_M2 + q8_M4) + q\
8_R4*( q8_M2 - <identity ...>) ),
  ( q8_Y16*( -<identity ...>), q8_R1*( -<identity ...>) + q8_R2*( -q8_M1) + q\
8_R4*( q8_M3 + q8_M1) ) ] ]
gap> RootIdentities( q8 );
[ ( q8_Y3*( q8_M1*q8_M4), q8_R1*( q8_M1 - <identity ...> ) ),
  ( q8_Y3*( q8_M1*q8_M4), q8_R1*( q8_M1 - <identity ...> ) ) ]

```

5.4.2 RootIdentities

◇ RootIdentities(*grp*)

(attribute)

The *root identities* are identities of the form $r^w r^{-1}$ where $r = w^n$ is a relator and $n > 1$.

Example

```

gap> RootIdentities( q8 );
[ ( q8_Y3*( q8_M1*q8_M4), q8_R1*( q8_M1 - <identity ...> ) ),
  ( q8_Y3*( q8_M1*q8_M4), q8_R1*( q8_M1 - <identity ...> ) ) ]

```

References

- [BH82] Ronald Brown and Johannes Huebschmann. Identities among relations. In R. Brown and T.L. Thickstun, editors, *Low-Dimensional Topology*, volume 46 of *London Math. Soc. Lecture Note Series*, pages 153–202. Cambridge University Press, 1982. [5](#)
- [BRS99] R. Brown and A. Razak Salleh. On the computation of identities among relations and of free crossed resolutions of groups. *London Math. Soc. J. Comput. Math.*, 2:28–61, 1999. [5](#)
- [Hey99] Anne Heyworth. *Applications of Rewriting Systems and Groebner Bases to Computing Kan Extensions and Identities Among Relations*. Ph.D. thesis, University of Wales, Bangor, 1999. [5](#)
- [HW03] Anne Heyworth and Christopher D. Wensley. Logged rewriting and identities among relators. In C. M. Campbell, E. F. Robertson, and G. C. Smith, editors, *Groups St Andrews 2001 in Oxford*, volume 304 of *London Math. Soc. Lecture Note Series*, pages 256–276. Cambridge University Press, 2003. [5](#)

Index

$=,+,*$ for module polynomials, [23](#)

$=,+,*$ for monoid polynomials, [18](#)

AddTermModulePoly, [23](#)

AddTermMonoidPoly, [18](#)

Coeffs, [17](#)

ElementsOfMonoidPresentation, [11](#)

FreeGroupOfPresentation, [7](#)

FreeRelatorGroup, [7](#)

FreeRelatorHomomorphism, [7](#)

GroupRelatorsOfPresentation, [7](#)

HomomorphismOfPresentation, [7](#)

IdentitiesAmongRelators, [24](#)

InverseRelatorsOfPresentation, [7](#)

KnuthBendix, [10](#)

LeadCoeffMonoidPoly, [17](#)

LeadMonoidPoly, [22](#)

LeadTerm, [17](#), [22](#)

Length, [19](#), [22](#)

LoggedKnuthBendix, [13](#)

LoggedOnePassKB, [12](#)

LoggedOnePassReduceWord, [14](#)

LoggedReduceWordKB, [14](#)

LoggedRewriteReduce, [13](#)

LoggedRewritingSystemFpGroup, [15](#)

ModulePoly, [21](#)

Monic, [17](#)

MonoidPoly, [16](#)

MonoidPolyFromCoeffsWords, [16](#)

MonoidPresentationFpGroup, [7](#)

One, [22](#)

OnePassKB, [10](#)

OnePassReduceWord, [9](#)

ReduceMonoidPoly, [19](#)

ReduceWordKB, [9](#)

RewriteReduce, [10](#)

RewritingSystemFpGroup, [8](#)

RootIdentities, [25](#)

ShorterLoggedRule, [14](#)

ShorterRule, [10](#)

Terms, [17](#), [22](#)

Words, [17](#)

ZeroModulePoly, [23](#)

ZeroMonoidPoly, [16](#)